NAG Library Routine Document

F08NEF (DGEHRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08NEF (DGEHRD) reduces a real general matrix to Hessenberg form.

2 Specification

```
SUBROUTINE F08NEF (N, ILO, IHI, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER            N, ILO, IHI, LDA, LWORK, INFO
REAL (KIND=nag_wp) A(LDA,*), TAU(*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name ***dgehrd***.

3 Description

F08NEF (DGEHRD) reduces a real general matrix A to upper Hessenberg form H by an orthogonal similarity transformation: $A = QHQ^{\mathrm{T}}$.

The matrix Q is not formed explicitly, but is represented as a product of elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 9).

The routine can take advantage of a previous call to F08NHF (DGEBAL), which may produce a matrix with the structure:

$$\begin{pmatrix} A_{11} & A_{12} & A_{13} \\ & A_{22} & A_{23} \\ & & A_{33} \end{pmatrix}$$

where A_{11} and A_{33} are upper triangular. If so, only the central diagonal block A_{22}, in rows and columns i_{lo} to i_{hi}, needs to be reduced to Hessenberg form (the blocks A_{12} and A_{23} will also be affected by the reduction). Therefore the values of i_{lo} and i_{hi} determined by F08NHF (DGEBAL) can be supplied to the routine directly. If F08NHF (DGEBAL) has not previously been called however, then i_{lo} must be set to 1 and i_{hi} to n.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: ILO – INTEGER *Input*

3: IHI – INTEGER *Input*

On entry: if A has been output by F08NHF (DGEBAL), then ILO and IHI **must** contain the values returned by that routine. Otherwise, ILO must be set to 1 and IHI to N.

Constraints:

if $N > 0$, $1 \leq ILO \leq IHI \leq N$;
if $N = 0$, $ILO = 1$ and $IHI = 0$.

4: A(LDA, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the n by n general matrix A.

On exit: A is overwritten by the upper Hessenberg matrix H and details of the orthogonal matrix Q.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08NEF (DGEHRD) is called.

Constraint: $LDA \geq \max(1, N)$.

6: TAU($*$) – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array TAU must be at least $\max(1, N - 1)$.

On exit: further details of the orthogonal matrix Q.

7: WORK($\max(1, LWORK)$) – REAL (KIND=nag_wp) array *Workspace*

On exit: if $INFO = 0$, $WORK(1)$ contains the minimum value of LWORK required for optimal performance.

8: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08NEF (DGEHRD) is called.

If $LWORK = -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, $LWORK \geq N \times nb$, where nb is the optimal ***block size***.

Constraint: $LWORK \geq \max(1, N)$ or $LWORK = -1$.

9: INFO – INTEGER *Output*

On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$INFO < 0$

If $INFO = -i$, argument i had an illegal value.

If $INFO = -999$, dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed Hessenberg matrix H is exactly similar to a nearby matrix $(A + E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n, and ϵ is the ***machine precision***.

The elements of H themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues, eigenvectors or Schur factorization.

8 Parallelism and Performance

F08NEF (DGEHRD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08NEF (DGEHRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}q^2(2q + 3n)$, where $q = i_{\text{hi}} - i_{\text{lo}}$; if $i_{\text{lo}} = 1$ and $i_{\text{hi}} = n$, the number is approximately $\frac{10}{3}n^3$.

To form the orthogonal matrix Q F08NEF (DGEHRD) may be followed by a call to F08NFF (DORGHR):

```
CALL DORGHR(N,ILO,IHI,A,LDA,TAU,WORK,LWORK,INFO)
```

To apply Q to an m by n real matrix C F08NEF (DGEHRD) may be followed by a call to F08NGF (DORMHR). For example,

```
CALL DORMHR('Left','No Transpose',M,N,ILO,IHI,A,LDA,TAU,C,LDC, &
             WORK,LWORK,INFO)
```

forms the matrix product QC.

The complex analogue of this routine is F08NSF (ZGEHRD).

10 Example

This example computes the upper Hessenberg form of the matrix A, where

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix}.$$

10.1 Program Text

```
    Program f08nefe

!     F08NEF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dgehrd, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
```

```
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter    :: zero = 0.0E0_nag_wp
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, lwork, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), tau(:), work(:)
!     .. Executable Statements ..
      Write (nout,*) 'F08NEF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      lwork = 64*n
      Allocate (a(lda,n),tau(n-1),work(lwork))

!     Read A from data file

      Read (nin,*)(a(i,1:n),i=1,n)

!     Reduce A to upper Hessenberg form

!     The NAG name equivalent of dgehrd is f08nef
      Call dgehrd(n,1,n,a,lda,tau,work,lwork,info)

!     Set the elements below the first subdiagonal to zero

      Do i = 1, n - 2
        a(i+2:n,i) = zero
      End Do

!     Print upper Hessenberg form

      Write (nout,*)
      Flush (nout)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,n,a,lda,'Upper Hessenberg form',ifail)

    End Program f08nefe
```

10.2 Program Data

```
F08NEF Example Program Data
  4                           :Value of N
  0.35   0.45  -0.14  -0.17
  0.09   0.07  -0.54   0.35
 -0.44  -0.33  -0.03   0.17
  0.25  -0.32  -0.13   0.11   :End of matrix A
```

10.3 Program Results

```
 F08NEF Example Program Results

 Upper Hessenberg form
          1       2       3       4
 1   0.3500 -0.1160 -0.3886 -0.2942
 2  -0.5140  0.1225  0.1004  0.1126
 3   0.0000  0.6443 -0.1357 -0.0977
 4   0.0000  0.0000  0.4262  0.1632
```